

Гилев А.И., - Соменов Л.Н.

Б2-10-86-246

УР405+УР452

3287/86



ОБЪЕДИНЕННЫЙ ИНСТИТУТ ЯДЕРНЫХ ИССЛЕДОВАНИЙ

Б 2 - 10 - 86 - 246

ДЕПОНИРОВАННАЯ ПУБЛИКАЦИЯ

Дубна 1986

# ДРАЙВЕР ИНТЕРФЕЙСА НМЛ В СТАНДАРТЕ КАМАК, ВКЛЮЧЕННЫЙ В СИСТЕМУ РАФОС

Гилев А.И., Сомов Л.Н.

## I. Введение

В настоящее время получили широкое распространение вычислительные комплексы на базе микро-ЭВМ типа "Электроника-60", ДВК и их аналогов. Стандартный набор внешних устройств таких комплексов состоит из алфавитно-цифрового дисплея, накопителя на гибких магнитных дисках и мозаичного печатающего устройства.

Для проведения небольших физических экспериментов применение такого типа ЭВМ вполне оправдано из-за относительно малой стоимости вычислительного комплекса. В то же время регистрируемую информацию необходимо записывать на накопители большей емкости, например, на магнитную ленту (МЛ). Однако указанные ЭВМ не имеют в своем составе интерфейсов накопителей на МЛ. В ЛЯИ был разработан и применяется интерфейс КИ ОЗІ, занимающий блок единичной ширины в крейте КАМАК и позволяющий использовать накопитель на магнитной ленте (НМЛ) ИЗОТ 5003 и ИЗОТ 5005 в составе вычислительных комплексов на основе микро-ЭВМ /1/.

Данная работа содержит краткое описание тех изменений, которые необходимо внести в драйвер МТ системы РАФОС-2 с тем, чтобы можно было использовать НМЛ ИЗОТ 5003 совместно с ДВК. Драйвер обеспечивает файловую структуру и осуществляет связь НМЛ с ЭВМ через интерфейс КИ ОЗІ и контроллер крейта в составе интерфейса магистрали КК 007 /2/ и интерфейса микро-ЭВМ по каналу прямого доступа КЭ 002 /2/. Блок КИ ОЗІ занимает 15-е место в крейте КАМАК.

## II. Описание блоков КИ ОЗІ, КК 007, КЭ 002

Подробное описание блоков КИ ОЗІ, КК 007 и КЭ 002 дано соответственно в работах /1/ и /2/. Здесь приводятся краткие сведения о работе блоков, необходимые для понимания работы драйвера.

Интерфейс КИ ОЗІ предназначен для использования НМЛ ИЗОТ 5003 и ИЗОТ 5005 в составе вычислительных комплексов при управлении через соответствующий контроллер ЭВМ. Блок КИ ОЗІ задает 8 операций НМЛ, которые приведены в таблице I.

Таблица I. Операции, задаваемые интерфейсом.

Операция	Нет опера- ции	Воспро- извед. вперед	Воспро- извед. назад	Запись	Запись контр. сумм	Запись марке- ра файла	Пере- мотка	Конец опера- ции
Обозна- чение	НОП	ВВ	ВН	ЗАП	ЗКС	ЗМФ	ПРМ	КОП
Код опера- ции	000	001	010	011	100	101	110	111

Запуск операции производится по команде F(17) в соответствии с кодом из таблицы I на шинах W9+W11.

Набор всех команд и процедуры их исполнения с помощью интерфейса КИ ОЗІ представлены в таблице II.

Таблица II. Команды управления НМЛ и последовательность их исполнения.

Команда	Старто- вая операция	Сигнал синхро- низации	Команды переда- чи дан- ных	Завершающие операции
Записать зону	ЗАП	L1	F(16)	ЗКС, КОП
Прочитать зону	ВВ	L1	F(0)	КОП
Записать маркер файла	ЗАП	L1	-	ЗМФ, КОП
Стереть промежуток	ЗАП	L1	-	ЗАП (9 раз), КОП
Пропустить зону вперед/назад	ВВ/ВН	L2	-	КОП
Пропустить файл вперед/назад	ВВ/ВН	L3	-	КОП
Перемотать	ПРМ	L4	-	-

Любая команда начинается стартовой операцией и заканчивается одной или несколькими завершающими операциями. При задании каждой операции командой F(17) необходимо разблокировать соответствующие сигналы L1+L4. Сигналом синхронизации всех завершающих операций является сигнал L1.

В таблице III показано назначение шин записи и чтения магистрали при использовании команд управления интерфейсом.

Таблица III. Назначение шин записи и чтения магистралей при использовании команд управления интерфейсом.

NAF	W	I5	I4	I3	I2	II	IO	9	8	7	6	5	4	3	2	I	0	
										Данные								
NA(O)F(16)										0	I	2	3	4	5	6	7	
NA(O)F(17)		Разблоки- ровка				Вы- бор НМЛ		Код операции										
		L4	L3	L2	L1													
NAF	R	I5	I4	I3	I2	II	IO	9	8	7	6	5	4	3	2	I	0	
										Данные								
NA(O)F(0)										K	0	I	2	3	4	5	6	7
NA(O)F(1)							ГТВ	ГТВ	НЗЗ	КД	КЛ	НЛ	ОШ	L4	L3	L2	L1	
							I	0					ЦКС					

где НЛ - начало ленты; КЛ - конец ленты, КД - конец данных, НЗЗ - нет защиты записи, ГТВО и ГТВЛ - готовность устройств.

Интерфейс магистралей КК 007 содержит регистр MNAF и обеспечивает 3 режима обмена между магистралью и ЭВМ:

M(0) - обмен одним словом;

M(2) - обмен массивом по одному адресу в режиме ULS;

M(3) - обмен массивом при последовательном опросе адресов в режиме АСА.

Код для выбора флага задается по пяти линиям ВФ:

ВФ(1)+ВФ(23) - запросы L1+L23;

ВФ(3I) - логическая "I".

Возможны два варианта начала обмена:

- при одновременном наличии сигнала "Начало связи" и флага;

- сразу после подачи сигнала "Начало связи" при ВФ = 3I.

Интерфейс микро-ЭВМ "Электроника-60" по КПД КЭ 002 обеспечивает обмен информацией в формате 16-разрядных слов или массива таких слов через канал прямого доступа (КПД) и предназначен для работы совместно с интерфейсом магистралей КК 007.

Программно доступны четыре 16-разрядных регистра, которые занимают четыре последовательных адреса, начиная от I77000:

- регистр управления и статуса (РУС), назначение всех 16 разрядов которого приведено в таблице IV;

- счетчик адреса (СА), указывающий адрес ячейки ОЗУ для пересылки очередного слова данных;

- регистр (счетчик) переданных слов (СС), работающий на вычитание;

- регистр MNAF, который находится в блоке КК 007 и назначение всех 16 разрядов которого приведено в таблице У.

Таблица ИУ. Назначение разрядов регистра управления и статуса (РУС).

15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
Наличие флага	Ошибка обращ. к памяти	Импульс оконч. связи	Линии выбора флага					Сост. тригг. связи	Разреш. прерыв.				0-зап. I-чтен.	Импульс начала связи	

Таблица У. Назначение разрядов регистра MNAF.

15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
M			F						N			A			

С программно доступными регистрами можно работать как с ячейками памяти. Обмен массивом начинается при записи 1 в нулевой разряд РУС. Окончание обмена может происходить по сигналу "Конец связи" от блока КК 007, при достижении "0" счетчиком слов, при обнаружении ошибки в обращении к памяти или по инициативе ЭВМ. Микро-ЭВМ определяет конец обмена по прерыванию с вектором V, если оно разрешено, или путем программной проверки 7-го разряда РУС.

### Ш. Фрагменты драйвера и их краткое описание

Общим для всех операций, выполняемых драйвером, является использование макрокоманд, обеспечивающих передачу данных и команд через магистраль КАМАК. Эти макрокоманды имеют следующий вид:

```
.MACRO   MOVADR   LOC,R
MOV      PC,R
ADD      #LOC-.,R
.ENDM    MOVADR
```

;-----

```
.MACRO   STBUF
ADD      NMG,R5
SWAB    R5
MOV      R5,WBUF
.ENDM    STBUF
```

;-----

```

.MACRO SETDIR F
.IF LT F
DIR = 0
.MEXIT
.ENDC
.IF LT F-8.
DIR = 1
.MEXIT
.ENDC
.IF LT F-16.
DIR = 0
.MEXIT
.ENDC
.IF LT F-24.
DIR = 3
.MEXIT
.ENDC
DIR = 0
.ENDM SETDIR

```

```

;-----
I1: .MACRO CAMWT ?I1
TSTB @#MTR
BPL I1
.ENDM CAMWT

```

```

;-----
.MACRO KE002 M,N,A,F,FLAG,BUF
SETDIR F
CAMWT
MOV #((M*40000)+(F*1000)+(N*20)+A),@#MTMNAF
MOVADR BUF,R2
MOV R2,@#MTADR
MOV #((FLAG*400)+DIR),@#MTSE.
CAMWT
.ENDM KE002

```

```

;-----
.MACRO EOP A
MOV #(EOPC+EL1),R5
ADD NMG,R5
SWAB R5
MOV R5,ABUF
CAMWT

```

KE002 0,NST,0,17.,A,ABUF  
.ENDM EOP

Следует отметить, что поскольку используется канал прямого доступа к ОЗУ, к которому могут быть подключены и другие внешние устройства, то более эффективным представляется применение режима M(2), т.е. обмен массивами данных по одному адресу.

Последнее общее замечание заключается в том, что в драйвере выделен промежуточный буфер ВВУФ, поскольку блок КИ ОЗИ осуществляет побайтную передачу данных, а контроллер - передачу 16-разрядными словами. Размер блока выбран равным 514, так как стандартный размер блока в системе равен 512 байтам. Если при работе с НМЛ требуется передача блоков с большей размерностью, то ВВУФ следует увеличить на необходимое число слов.

1. Операция чтения REDEF обеспечивает передачу блока данных с максимальной размерностью, равной 512 байтам с МЛ в ОЗУ. Операция прекращается при появлении сигнала L2 (конец зоны). Собственно чтение данных прекращается не по достижению счетчиком слов контроллера значения 0, а по  $Q = 0$ . Поэтому при передаче блоков большей размерности необходимо в счетчик слов загружать число со значением большим, чем длина блока. Вид REDEF приводится ниже.

```
REDEF:  NOP
        STBUF
        CLR     MTS
        CLR     MTC
        MOV     R1,CNR
        MOV     #-514.,R5
        MOVADR  BBUF,R1
5$:     MOV     #0,(R1)+
        INC     R5
        BMI     5$
        MOV     -(R3),R1      ; B'JFF.ADDR.
        KE002  0,NST,0,17.,NST,WBUF
        CAMWT
        MOV     #514.,@ MTCNR
        KE002  2,NST,0,0,NST,BBUF
        CAMWT
        MOV     @#MTCNR,CC
```

```

6$: KE002      0,NST,0,1,NST,BUF
   BIT        #2,BUF                ; EOZ ?
   BEQ        6$                    ; NO
   EOP        NST
   BIT        #4,BUF                ; EOF ?
   BEQ        7$                    ; NO
   BIS        #40000,MTS
   BIS        #-1,MTC
   JMP        EXIN
7$:  NEG        CC
   ADD        #514.,CC
   CMP        CC,CNR                ; L-ZONE - L-REQ.?
   BEQ        9$                    ; L-ZONE = L-REQ
   BLT        10$
   BIS        #RLE,MTS              ; L-ZONE L-REQ.
   BIS        #-1,MTC              ;
   BR         9$                   ;
10$: BIS        #BTEOPI,MTS        ; L-ZONE L-REQ.
   BIS        #-1,MTC              ;
   JMP        EXIN                 ;
9$:  NEG        CNR
   ADD        CNR,CC               ; L-ZONE - L-REQ.
   MOV        CNR,R5
   MOVADR     BBUF,R2
1$:  MOVB       (R2)+,(R1)+
   INC        R2
   INC        R5
   BMI        1$
2$:  BIT        #20,BUF            ; CRE ?
   BEQ        3$
   BIS        #-1,MTC
   BIS        #20000,MTS
3$:  BIT        #100,BUF           ; EOT ?
   BEQ        4$                   ; NO
   BIS        #-1,MTC
   BIS        #2000,MTS
4$:  JMP        EXIN

```

2. Операция записи WRITE обеспечивает передачу блока данных максимальной размерностью, равной 512 байтам из ОЗУ на МЛ. Здесь длина блока ограничивается только размером BBUF. Вид WRITE приводится ниже.



```

WRITE:  NOP
        STBUF
        KE002    0,NST,0,1,31.,BUF
        BIT      #400,BUF          ;WRITE LOCK ?
        BNE      2$                ; NO

        MOV      #-1,MTC
        MOV      #4,MTS
        JMP      EXIN

2$:     BIT      #40,BUF          ; BOT ?
        BEQ      5$                ; NO
        MOV      #177700,RO

4$:     KE002    0,NST,0,17.,NST,WBUF
        INC      RO
        BMI      4$

5$:     MOV      #177774,RO
1$:     KE002    0,NST,0,17.,NST,WBUF
        INC      RO
        BMI      1$

RIGHT1: CLR      MTS
        CLR      MTC
        MOV      R1,R5
        DEC      R1
        CAMWT
        MOV      R1,@#MTCNR
        MOV      -(R3),R4
        MOVADR   BBUF,R2
        NEG      R5

1$:     MOVB     (R4)+,(R2)+
        MOVB     #0,(R2)+
        INC      R5
        BMI      1$
        KE002    2,NST,0,16.,NST,BBUF
        MOVB     #(WRL+EL1),R5      ;WRITE CONTR.SUM.COMM.
        STBUF
        KE002    0,NST,0,17.,NST,WBUF
        KE002    0,NST,0,1,NST,BUF  ; READ STATUS MT
        EOP      NST
        BIT      #100,BUF          ; EOT ?
        BEQ      3$
        BIS      #-1,MTC
        BIS      #2000,MTS

3$:     JMP      EXIN

```

3. Операции записи с расширенным промежутком WREX, перемотки RWD, записи маркера файла WREOF, пропуска назад SKIPB и пропуска вперед SKIPF не нуждаются в комментариях и приводятся ниже.

```
WREX:  KE002    0,NST,0,1,31.,BUF
        BIT     #400,BUF                ;WRITE LOCK ?
        BNE     1$
        MOV     #-1,MTC
        MOV     #4,MTS
        JMP     EXIN
1$:     .REPT   9.
        KE002    0,NST,0,17.,31., WBUF
        .ENDR
        JMP     RIGHT1
```

-----

; REWIND

```
RWD:   CLR     MTS
        CLR     MTC
        KE002    0,NST,0,1,31.,BUF
        BIT     #40,BUF                ; BOT ?
        BEQ     6$                    ; NO.START REW.OPER.
        BIS     #40,MTS
        BR      7$
6$:     STBUF
        KE002    0,NST,0,17.,NST,WBUF
1$:     KE002    0,NST,0,1,31.,BUF
        BIT     #40,BUF                ; BOT ?
        BEQ     1$                    ; NO
        BIS     #40,MTS
7$:     EOP     NST
        JMP     EXIN
```

-----

; WRITE EOF

```
WREOF: NOP
        CLR     MTS
        CLR     MTC
        KE002    0,NST,0,1,31.,BUF
        BIT     #400,BUF                ; WRITE LOCK ?
        BNE     1$
        BIS     #-1,MTC
        BIS     #4,MTS
        JMP     EXIN
```

```

1$: STBUF
   MOV     #-12,R1
4$: KE002  0,NST,0,17.,NST,WBUF
   CAMWT
   INC     R1
   BMI     4$
   MOVB    #(FWRTTM+EL1),R5
   STBUF
   KE002  0,NST,0,17.,NST,WBUF
   KE002  0,NST,0,1,31.,BUF
   EOP     NST
   BIT     #100,BUF           ; EOT ?
   BEQ     2$                 ; NO
   BIS     #-1,MTC
   BIS     #2000,MTS
2$: BIT     #4,BUF           ; EOF ?
   BEQ     3$                 ; NO
   BIS     #40000,MTS
   BIS     #-1,MTC
3$: JMP     EXIN

```

```

;-----
SKIPB: NOP
      CLR     MTS
      CLR     MTC
      KE002  0,NST,0,1,31.,BUF
      BIT     #40,BUF
      BNE     4$
      ASR     R1
      TST     R1
      BMI     2$
      NEG     R1
      MOV     R1,CNR
      STBUF
      KE002  0,NST,0,17.,NST,WBUF
1$: KE002  0,NST,0,1,31.,BUF
   BIT     #42,BUF
   BEQ     1$
   BR      4$
2$: MOV     R1,CNR
   MOV     #(FSPBAK+EL3),R5
   STBUF
   KE002  0,NST,0,17.,NST,WBUF

```

```

3$:  KEOO2    0,NST,0,1,31.,BUF
      BIT     #44,BUF
      BEQ     3$
4$:  EOP      NST
      BIT     #4,BUF           ;EOF ?
      BEQ     5$
      BIS     #40000,MTS
      BIS     #-1,MTC
      JMP     EXIN
5$:  BIT     #40,BUF           ;BOT ?
      BEQ     6$
      BIS     #40,MTS
6$:  JMP     EXIN

```

-----

```

SKIPF:  NOP
        CLR     MTS
        CLR     MTC
        ASR     R1
        TST     R1
        BMI     2$
        NEG     R1
        MOV     R1,CNR
        STBUF
        KEOO2   0,NST,0,17.,NST,WBUF
1$:  KEOO2   0,NST,0,1,31.,BUF
      BIT     #2,BUF
      BR      4$
2$:  MOV     R1,CNR
      MOV     #(FSPFOR+EL3),R5
      STBUF
        KEOO2   0,NST,0,17.,NST,WBUF
3$:  KEOO2   0,NST,0,1,31.,BUF
      BIT     #4,BUF
      BEQ     3$
4$:  EOP      NST
      BIT     #4,BUF
      BEQ     5$
      BIS     #40000,MTS
      BIS     #-1,MTC
      JMP     EXIN

```

5\$: BIT #100, BUF ; FOT ?  
BEQ 6\$  
BIS #-1, MTC  
BIS #2000, MTS  
6\$: JMP EXIN

;-----  
EXIN: NOP  
JSR PC, RESTOR  
MOV #17500, @#MTR ; CAUSE AN INTERP.  
RTS PC

В заключение авторы выражают благодарность В.Т.Сидорову и И.Н.Чурину за полезные советы.

#### Литература

1. Ле Зон Пхир, В.Т.Сидоров. ОИЯИ, IO-8I-5I7, Дубна, I98I.
2. Вьонг Дао Ви и др. ОИЯИ, IO-8I-755, Дубна, I98I.
3. Е.А.Паскк и др. ОИЯИ, IO-83-347, Дубна, I983.

